Math 2280-1
Fri 4/21

(Final HW assignment is on Wed 4/19 notes)

Heat eqtn for temperature $u(x,t)$

$$u_t = k u_{xx}$$

On Wed. we found all product sol'ns
$$u(x,t) = X(x)\,T(t)$$

$$\left\{ \begin{array}{l} X T'(t) = k X''(x) T \\ \dfrac{T'}{kT} = \dfrac{X''}{X} = -\lambda \end{array} \right.$$

$\uparrow$ fcn of t, $\uparrow$ fcn of x

linear combos of

$$\underbrace{\underbrace{1, x}_{\lambda = 0}, \underbrace{\cos(rx)e^{-kr^2t}, \sin(rx)e^{-kr^2t}}_{\lambda = -r^2}}_{\text{suitable for IBVP's (1), (2)}}$$

$$\underbrace{\cosh(rx)\, e^{kr^2t}, \sinh rx\, e^{kr^2t}}_{\lambda = +r^2}$$

General Strategy to solve IBVP for heat eqtn:

(1) Get Fourier series for initial temperature $f(x)$ on interval $0<x<L$
- sine series if $u(0,t) = u(L,t) = 0$
- cosine series if $\frac{\partial u}{\partial x}(0,t) = \frac{\partial u}{\partial x}(0,L) = 0$

(2) For each
$a_n \cos\left(\frac{n\pi x}{L}\right)$ in cosine series, get $u_n(x,t) = a_n\left(\cos\frac{n\pi x}{L}\right)e^{-\frac{kn^2\pi^2}{L^2}t}$
$b_n \sin\left(\frac{n\pi x}{L}\right)$ in sine series, get $u_n(x,t) = b_n\left(\sin\frac{n\pi x}{L}\right)e^{-\frac{kn^2\pi^2}{L^2}t}$

general solution to IBVP will be (infinite) superposition of these sol'ns!

Example 2 page 613

$$\left\{ \begin{array}{l} u_t = k u_{xx} \\ u(0,t) = 0 \\ u(L,t) = 0 \\ u(x,0) = 100^\circ \end{array} \right. = f(x) \text{ for sine series}$$

$k = .15$ (iron)
$k = .005$ (concrete)
$L = 50$ cm



our old friend the square wave, but we just care about $0<x<50$

$$f \sim 100 \left[\frac{4}{\pi} \sum_{n \text{ odd}} \frac{1}{n} \sin\left(\frac{n\pi}{L} x\right) \right]$$ ← we are changing from t to x!

$$= \frac{400}{\pi} \sum_{n \text{ odd}} \frac{1}{n} \sin\left(\frac{n\pi}{50} x\right)$$

$$\Longrightarrow \quad \boxed{u(x,t) = \frac{400}{\pi} \sum_{n \text{ odd}} \frac{1}{n} \sin\left(\frac{n\pi}{50} x\right) e^{-k\left(\frac{n\pi}{50}\right)^2 t}}$$

<u>Notes</u>: n big $\Rightarrow$ exponential terms die off quickly in t. t large $\Rightarrow$ $n=1$ term dominates for $t \gg 1$

also, $u(x,t)$ always has max value at $x = 25$ (for t fixed); you can almost see this from the series!

Now go to Maple Handout

Example 3 page 617

insulated ends.



this is a multiple of our tent function, with half period $\underline{\underline{L/2}}$. now use cheat sheet

$$\text{tent}(x) \sim \frac{L}{4} - \frac{4(L/2)}{\pi^2} \sum_{n \text{ odd}} \frac{1}{n^2} \cos\left(\frac{n\pi x}{L/2}\right)$$

$$= \frac{L}{4} - \frac{2L}{\pi^2} \sum_{n \text{ odd}} \frac{1}{n^2} \cos\left(\frac{n\pi x}{L/2}\right)$$

$L = 50$, tent$(25) = 25$, so

$$f = 4\,\text{tent}(x) \sim 50 - \frac{400}{\pi^2} \sum_{n \text{ odd}} \frac{1}{n^2} \cos\left(\frac{n\pi x}{25}\right)$$

so

$$\boxed{u(x,t) = 50 - \frac{400}{\pi^2} \sum_{n \text{ odd}} \frac{1}{n^2} \cos\left(\frac{n\pi x}{25}\right) e^{-\left(\frac{n\pi}{25}\right)^2 k t}}$$

Same as text! book wrote it silly.

Fourier Series cheat sheet: (p. 593 + taking derivs)

(1)



$$f(t) = \frac{L}{2} - \frac{4L}{\pi^2} \sum_{n \text{ odd}} \cos\left(\frac{n\pi t}{L}\right) / n^2$$

(2)



$$f'(t) \sim \frac{4}{\pi} \sum_{n \text{ odd}} \frac{1}{n} \sin\left(\frac{n\pi t}{L}\right)$$

(3)



$$g(t) \sim \frac{2L}{\pi} \sum_{n} (-1)^{n+1} \frac{1}{n} \sin\left(\frac{n\pi t}{L}\right)$$

(4)



$$2\int_0^t g(\tau)\,d\tau = \frac{4L}{\pi^2} \sum_{n} (-1)^n \frac{1}{n^2} \cos\left(\frac{n\pi t}{L}\right)$$

$(= t^2 \text{ if } |t| \le L)$

Friday April 21
Fourier Series and the heat equation

General set up:

```
> restart:with(plots):
> a:=n->2/L*int(f(x)*cos(Pi*n/L*x),x=0..L):
    #for a cosine series.  We use x for the space variable,
    #so that we have t for time
  b:=n->2/L*int(f(x)*sin(Pi*n/L*x),x=0..L):
> cossum:=a(0)/2+sum(a(n)*cos(Pi*n/L*x),n=1..N):
    #cosine series
  sinsum:=sum(b(n)*sin(Pi*n/L*x),n=1..N):
    #sine series
> heatsol1:=sum(b(n)*sin(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1..N):
    #boundary conditions u(0,t)=u(L,t)=0
  heatsol2:=a(0)/2+sum(a(n)*cos(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1
  ..N):
    #boundary conditions for insulated ends
```

Example 2 page 613:

```
> f:=x->100:
  L:=50:
  N:=50:  #square wave is hard to approximate with Fourier, take big
  N
  k:=.15:  #thermal diffusivity for iron
>
> plot3d(heatsol1,x=0..50,t=0..1800,color=black,style=wireframe,
  axes=boxed,title='Figure 9.5.4');
```



Computer check of computations on page 614:
(a) temperature at midpoint after half an hour

```
> evalf(subs({t=1800,x=25},heatsol1));
```

43.84897699

(b) temperature for concrete:

```
> k:=.005:
  heatsol1:=sum(b(n)*sin(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1..N):
    #boundary conditions u(0,t)=u(L,t)=0
  evalf(subs({t=60*30,x=25},heatsol1));
  #half an hour later
```

99.99999917

```
> evalf(subs({t=60*60,x=25},heatsol1)); #one hour later
  evalf(subs({t=6*60*60,x=25},heatsol1));#six hours later
```

99.99381824

82.21276660

How many of you live in brick houses?
We could siee what the temperature profile looks like throughout the concrete slab at this last time value:

```
> slice:=z->evalf(subs({t=6*60*60,x=z},heatsol1));
```

$slice := z \rightarrow \mathrm{evalf}(\mathrm{subs}(\{t = 21600, x = z\}, heatsol1))$

```
> plot(slice(x),x=0..50,color=black, title='concrete temperature
  six hours later');
```



Example 3 page 616-617:

```
> restart:with(plots):
  a:=n->2/L*int(f(x)*cos(Pi*n/L*x),x=0..L):
    #for a cosine series.  We use x for the space variable,
    #so that we have t for time
  b:=n->2/L*int(f(x)*sin(Pi*n/L*x),x=0..L):
  cossum:=a(0)/2+sum(a(n)*cos(Pi*n/L*x),n=1..N):
    #cosine series
  sinsum:=sum(b(n)*sin(Pi*n/L*x),n=1..N):
    #sine series
  heatsol1:=sum(b(n)*sin(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1..N):
    #boundary conditions u(0,t)=u(L,t)=0
  heatsol2:=a(0)/2+sum(a(n)*cos(Pi*n/L*x)*exp(-n^2*Pi^2*k*t/L^2),n=1
  ..N):
    #boundary conditions for insulated ends

>
  L:=50:
  N:=10:
  k:=.15:  #thermal diffusivity for iron
  f:=x->100-4*abs(x-25);


  plot(f(x),x=0..50,color=black);
```

$$f := x \rightarrow 100 - 4\,|x - 25|$$




```
> plot3d(heatsol2,x=0..50,t=0..1200,color=black,
  style=wireframe, axes=boxed, title='Figure 9.5.6');
```




```
> Digits:=5;
  cossum;
  heatsol2;
```

$$Digits := 5$$

$$50 - \frac{400\cos\left(\frac{\pi x}{25}\right)}{\pi^2} - \frac{400}{9}\,\frac{\cos\left(\frac{3\,\pi x}{25}\right)}{\pi^2} - \frac{16\cos\left(\frac{\pi x}{5}\right)}{\pi^2}$$

$$50 - 40.528\cos(0.12566\,x)\,\mathbf{e}^{(-0.0023687\,t)} - 4.5031\cos(0.37699\,x)\,\mathbf{e}^{(-0.021319\,t)}$$
$$- 1.6211\cos(0.62832\,x)\,\mathbf{e}^{(-0.059218\,t)}$$

\>